

BRITISH OCCUPATIONAL HYGIENE SOCIETY

*HYDROCARBON DISTILLATE
VAPOUR COMPOSITION
- Prediction by Microcomputer*

Technical Guide Series, No. 2

October 1983

TECHNICAL GUIDE SERIES

- No. 1 Statistical Analysis of Monitoring Data by Microcomputer. August 1983
No. 2 Hydrocarbon Distillate Vapour Composition – Prediction by Microcomputer. October 1983

Copies of the Technical Guide series are obtainable price £2.00 post free (cash with order) from the British Occupational Hygiene Society, Honorary Business Secretary, Mr P.J. Hewitt, University of Bradford. Bradford BD7 1DP

HYDROCARBON DISTILLATE VAPOUR COMPOSITION -
PREDICTION BY MICROCOMPUTER

by D.T. Coker

Esso Europe Inc,
Industrial Hygiene Unit, Esso Research Centre,
Abingdon, Oxfordshire OX13 6AE.

A Basic programme for calculating the composition and concentration of vapours from the surface of bulk hydrocarbon distillate liquids.

British Occupational Hygiene Society Technical Guide No. 2.

The BOHS Technical Guide series is intended for reports of practical use in occupational hygiene, mainly produced by the BOHS Technology Committee. The responsibility for accuracy and for opinions stated remains with the author, and not with the Society.

British Occupational Hygiene Society,
London School of Hygiene and Tropical Medicine,
Keppel Street, London WC1E 7HT

1983

HYDROCARBON DISTILLATE VAPOUR COMPOSITION -
PREDICTION BY MICROCOMPUTER

by D.T. Coker

This is the second paper in the series from the BOHS Technology Committee giving applications and programmes for microcomputers which may be of use to the occupational hygienist. The Committee would welcome any programmes from hygienists to include in this series, covering any aspect of occupational hygiene.

The application and programme given here is for predicting the composition and concentration of vapours arising from evaporation from the surface of bulk hydrocarbon distillate liquids. The vapour composition is typically very different to the liquid, but can be predicted using Raoult's Law for perfect gas behaviour and equilibrium conditions.

This prediction gives a reasonable estimate of the vapour composition, but if better accuracy is required then more complex calculations involving activity coefficients must be used - such as in the American Petroleum Institute's VAPRAT programme.

Hydrocarbons behave as near perfect gases and conditions close to equilibrium occur when evaporation is from a bulk liquid into a still, enclosed headspace, at a steady temperature and pressure. However a different composition - closer to the liquid composition - will occur when evaporation is from a thin film such as a wet surface. The range of vapour compositions possible therefore is limited between the liquid composition and the predicted composition.

Raoult's Law states that:

$$P_c = f_c \times S_c \quad \xrightarrow{1}$$

where P_c = partial vapour pressure of the vapour from a component c of a mixture

f_c = mole fraction of c in the mixture

S_c = Saturated Vapour Pressure (SVP)

similarly $P = \Sigma (f \times S)$

where P = total vapour pressure of the mixture in mmHg and

$= \Sigma (f \times S) =$ sum of $(f \times S)$ for each component of the mixture

The mole fractions of mixture components can be calculated from their weight percent in the mixture and their molecular weights.

Calculating Vapour Pressure

SVP's for hydrocarbons can be read from the graph (Figure 1). However for computing it is more convenient to use a mathematical expression⁽¹⁾. The following expression gives the SVP's for hydrocarbons from C No.4 and above with a good degree of accuracy.

$$P_C = 10^{\left(\frac{2.881 - \phi \Delta t}{273 + t - 0.15 \Delta t} \right) - 2}$$

where t = temperature for the calculation in $^{\circ}\text{C}$

Δt = difference between t and the boiling point of component C

P_C = SVP of component C (in mmHg)

ϕ = Entropy of Vapourisation
(B.Pt $^{\circ}\text{C}$ \times 0.00205 + 4.54 for hydrocarbons) — 3

This does not give an accurate value for C 1 to 3 hydrocarbons and a separate expression is required for each. (These are given in the programme).

Component Grouping

It is obviously rather impractical to calculate the composition of all the many hundreds of possible components in a distillate, but a convenient condensation can be made by grouping as follows:

- saturates + olefines by carbon number
- aromatics by carbon number

As the SVP's are very low above carbon number 13 this is used as the upper limit, any components of higher carbon number can be included with C. No. 13.

The composition of the vapour is often used for calculating Exposure Limits and so certain key substances with individual exposure limits are included separately. These are:

cyclohexane
 nhexane
 trimethylbenzenes
 isopropylbenzene
 naphthalene
 benzene (as C₆ aromatic)

This composition can be obtained by a standard GC analysis.

The boiling point of each carbon number group of saturates is taken as the n-paraffin and in the second group as a mean figure for the most common alkylbenzenes. These boiling points and their calculated ϕ values and molecular weights are given in Table 1.

Saturates + Olefines				Aromatics			
C No.	BP ^o C	ϕ	M.W.	C No.	BP ^o C	ϕ	M.W.
1	-164	4.20	16	6	80	4.70	78
2	- 88	4.36	30	7	111	4.77	92
3	- 42	4.45	44	8	140	4.83	106
4	- 1	4.54	58	9	165	4.88	120
5	36	4.61	72	10	185	4.92	134
6	69	4.68	86	11	205	4.96	148
7	98	4.74	100	12	226	5.01	162
8	126	4.80	114	13	245	5.05	176
9	151	4.85	128	Trimethyl Bz	170	4.89	120
10	174	4.90	142	Isopropyl Bz	152	4.85	120
11	196	4.94	156	Naphthalene	218	4.99	128
12	216	4.99	170	Cyclohexane	81	4.71	84
13	236	5.03	184	n-Hexane	69	4.68	86

An alternative to using the carbon number analysis above is to estimate total carbon number distribution from the distillation curve, using the above boiling points or a GC distillation and an aromatics GC analysis. Aromatics are then subtracted from the relevant carbon number.

Calculation Procedure

The vapour composition can now be calculated from liquid compositional analysis in a series of steps as shown in the example in the following tabulation.

Group	Wt %	Mole fraction (wt % / MW)	Normalised Mole fraction	Calculated SVP @ 20°C (mmHg)	Partial	Vol % Composition C/7.60
					Pressure A X B (mmHg)	
C ₄	10	.172	.144	1596	230	30.3
C ₅	20	.278	.232	424	98.4	12.9
C ₆	20	.233	.194	120	23.3	3.1
C ₇	15	.150	.125	36.8	4.60	0.61
C ₈	5	.044	.037	11.1	.41	0.05
C ₆ A	5	.064	.053	77.3	4.10	0.54
C ₇ A	15	.163	.136	21.2	2.88	0.38
C ₈ A	10	.094	.078	5.9	.46	0.06
Σ = 100		Σ = 1.198	Σ = 1	Σ 364.2		
Total vapour pressure						47.9

Step 1 Divide the weight percent of each group by its molecular weight

Step 2 Normalise these to give a sum of 1
 This is the molar fraction of the group (A)

- Step 3 Calculate the Saturated Vapour Pressures for each group at the required temperature from equations (2) and (3) or from Figure 1. The units will be in mmHg (B)
- Step 4 Calculate the partial pressures of each group by multiplying the molar fractions by the SVP's (A X B = C) units will be mmHg
- Step 5 Sum the partial pressures for each group to give the Total Vapour Pressure in mmHg
- Step 6 Calculate the volume percent composition of the vapour by dividing by 7.60, or multiply by $\frac{100}{\text{Total Vapour Pressure}}$ to give composition normalised to 100% (i.e. excluding air).

Computer Programme

A Computer programme is given in Appendix 1 written in Microsoft Basic for a Commodore PET 2001. This inputs liquid composition for the component groups and calculates vapour composition at the required temperature. It also calculates the average molecular weight for each composition and the total % hydrocarbon vapour. The calculations can then be repeated for any temperature.

Some alterations may be necessary for other variants of Basic such as "LET" and "RESTORE" commands, and these will need to be checked by the user to ensure the calculations are properly executed. The tabulation of results starting line 800 may also need to be TABbed in some computers.

The input composition must add up to 99 to 101%. The values for the 5 individual substances must also be included in their respective carbon number groups, i.e. n-Hexane and Cyclohexane in Sats. C. No. 6; Trimethylbenzenes and isopropylbenzene in Aromatics C. No. 9; Naphthalene in Aromatics C. No. 10.

COMPOSITION OF TEST SET

<u>COMPONENT</u>	<u>LIQUID WT %</u>	<u>VAPOUR @ 13°C VOL %</u>	<u>MEASURED VAPOUR @ 13°C VOL %</u>
SATS. C No 1	0	0	0
SATS. C No 2	0	0	0
SATS. C No 3	.3	12.24	11.2
SATS. C No 4	7.6	56.33	53.5
SATS. C No 5	14.8	22.4	25.0
SATS. C No 6	18.4	6.29	7.5
SATS. C No 7	7.8	.68	1.2
SATS. C No 8	4.7	.1	0.2
SATS. C No 9	1.7	.01	.02
SATS. C No 10	2.7	0	0
SATS. C No 11	.89	0	0
SATS. C No 12	0	0	0
SATS. C No 13	0	0	0
AROMS C No 6	4.5	1.08	0.9
AROMS C No 7	12.7	.68	.49
AROMS C No 8	14.1	.17	.1
AROMS C No 9	5.9	.02	.01
AROMS C No 10	3.9	0	0
AROMS C No 11	0	0	0
AROMS C No 12	0	0	0
AROMS C No 13	0	0	0
CYCLOHEXANE	0	0	0
N-HEXANE	3.5	1.2	1.5
TRIMETHYLBENZENES	3.9	.01	0
ISOPROPYLBENZENE	0	0	0
NAPHTHALENE	0	0	0
BENZENE	4.5	1.08	0.9
AVE. MOL WTS:	94	62.1	
TOTAL %VOL H/C IN VAPOUR	34.461		

PETVAP PROGRAMME SUMMARY

Line Nos.	Function
10 - 40	Heading
110 - 400	Inputs liquid composition, calculates mol fractions, sums input percentages and mol fractions
430 - 480	Checks liquid composition adds up to 99 to 101%
520 & 700	Boiling points of components (Sats and Aromatics)
530 - 776	Calculates partial pressure of the following components and sums total partial pressure
590	Methane
610	C ₂
630	C ₃
660	Saturates C ₄ to C ₁₃
740	Aromatics C ₆ to C ₁₃
775	The 5 individual hydrocarbons
790	Left free for line printer output commands
800 - 820	Output table headings
830 - 950	Prints output table of: Component:% wt in liquid: % vol in vapour (normalised to 100%)
970 - 1110	Calculates and prints average molecular weight of liquid and vapour, and total vapour concentration
1130	Is left free for line printer termination commands
1140 - 1200	Allows the procedure to be repeated for another temperature

PROGRAMME LISTING

```


10 PRINT"PETVAP - THIS PROGRAMME PREDICTS VAPOUR"
20 PRINT"COMPOSITION FROM LIQUID COMPOSITION FOR"
30 PRINT"PETROLEUM DISTILLATES.IT ALSO CALCULATES"
40 PRINT"OEL'S,AVERAGE MOL WTS AND %VAPOUR IN AIR"
50 PRINT
60 PRINT"ENTER NAME OF PRODUCT"
70 INPUT A$
80 PRINT
90 X=0:Z=0:G=0:E1=0:E2=0:N=6
100 DIM A(13),B(13),C(13),D(13),K(N),J(N),H(13),L(13),R(N)
110 PRINT"ENTER LIQUID COMPOSITION IN WT%"
120 FOR I=1TO13
130 PRINT"SATS. C NO";I;
140 INPUT A(I)
150 H(I)=A(I)/(2+I*14)
160 X=X+A(I)
170 Z=Z+H(I)
180 NEXT I
270 FOR I=6TO13
280 PRINT"AROMATICS C NO";I;
290 INPUT C(I)
300 L(I)=C(I)/(14*I-6)
310 X=X+C(I)
320 Z=Z+L(I)
330 NEXT I
340 DATA 84,"CYCLOHEXANE",86,"N-HEXANE",120,"TRIMETHYLBENZENES"
345 DATA120,"ISOPROPYLBENZENE",128,"NAPHTHALENES"
350 FOR I=1TO(N-1)
360 READ P,Q$
370 PRINT Q$;
380 INPUT J(I)
390 R(I)=J(I)/P
400 NEXT I
430 IF X<99 THEN460
440 IF X>101 THEN460
450 GOTO490
460 PRINT"TOTAL IS";X;".IT MUST BE BETWEEN 99 AND 101"
470 PRINT"TRY AGAIN"
475 RESTORE
477 X=0:Z=0
480 GOTO 110
490 PRINT
500 PRINT"ENTER TEMPERATURE IN DEG. C";
510 INPUT J
520 DATA -1,36,69,98,126,151,174,196,216,236
530 FOR I=1TO13
540 B(I)=H(I)/Z
550 IF I>3 THEN650
560 IF I=1 THEN590
570 IF I=2 THEN610
580 IF I=3 THEN630
590 B(I)=10*(6.89604-(449.746/(J+273)))*B(I)/7.6
600 GOTO 670
610 B(I)=10*(7.36795-(835.928/(J+273)))*B(I)/7.6
620 GOTO 670
630 B(I)=10*(7.18659-(990.853/(J+273)))*B(I)/7.6
640 GOTO 670
650 READ Q

```

```

660 B(I)=10↑(2.8808-(((Q*.00205+4.54)*(Q-J))/((273+J)-.15*(Q-J))))*B(I)/7.6
670 G=G+B(I)
690 NEXT I
700 DATA 80,111,140,165,185,199,226,245
710 FOR I=6T013
720 D(I)=L(I)/Z
730 READ Q
740 D(I)=10↑(2.8808-(((Q*.00205+4.54)*(Q-J))/((273+J)-.15*(Q-J))))*D(I)/7.6
750 G=G+D(I)
770 NEXT I
771 DATA 81,69,170,152,218
772 FOR I=1T0(N-1)
773 K(I)=R(I)/Z
774 READ Q
775 K(I)=10↑(2.8808-(((Q*.00205+4.54)*(Q-J))/((273+J)-.15*(Q-J))))*K(I)/7.6
776 NEXT I
780 PRINT
800 PRINT,,"COMPOSITION OF ";A$
810 PRINT,,"LIQUID","VAPOUR AT";J;"DEG C"
820 PRINT"COMPONENT","WT%", "VOL%"
830 PRINT"-----"
840 FOR I=1T013
850 PRINT"SATS. C NO";I,A(I),INT((B(I)*100/G)*100+.5)/100
860 NEXT I
890 FOR I=6T013
900 PRINT"AROMS. C NO";I,C(I),INT((D(I)*100/G)*100+.5)/100
910 NEXT I
920 DATA"CYCLOHEXANE","N-HEXANE","TRIMETHYLBENZENES","ISOPROPYLBENZENE"
925 DATA"NAPHTHALENE","BENZENE"
927 K(6)=D(6):J(6)=C(6)
930 FOR I=1TON
933 READ Q$
935 PRINT Q$,J(I),INT((K(I)*100/G)*100+.5)/100
940 NEXT I
950 PRINT"-----"
960 PRINT
970 FOR I=1T013
980 E1=E1+(A(I)/100)*(2+I*14)
990 E2=E2+(B(I)/G)*(2+I*14)
1000 NEXT I
1030 FOR I=6T013
1040 E1=E1+(C(I)/100)*(14*I-6)
1050 E2=E2+(D(I)/G)*(14*I-6)
1060 NEXT I
1090 PRINT"AVE. MOL WTS:",INT(E1*10+.5)/10,INT(E2*10+.5)/10
1100 PRINT
1110 PRINT"TOTAL %VOL H/C IN VAPOUR";INT(G*1000+.5)/1000
1120 PRINT
1140 PRINT"ENTER NEXT TEMPERATURE"
1150 RESTORE
1160 G=0:E1=0:E2=0
1170 FOR I=1TON-1
1180 READ Q,P$
1190 NEXT I
1200 GOTO 510
1210 END

```



Log 5 cycles x 2 mm. (multiplier)

Graph Data Ser. 8552